

Lesson 11


Sébastien Mathier

www.excel-pratique.com/en


Using Excel functions :

It is also possible to use Excel functions in VBA code, as we will see on this page.

After we have entered **WorksheetFunction**, the list of functions will appear :



Choose the function that you are interested in using, for example, COUNTBLANK :



In the following macro, the number of empty cells in the "A1:D8" range is stored in a variable and then displayed in a dialog box :

```
Sub test()  
    var_test = WorksheetFunction.CountBlank(Range("A1:D8"))  
    MsgBox var_test  
End Sub
```

Create a custom function :

In this example, we will create a custom IF function that can be used on a calculation worksheet like any other function.

To display a value (B2) if a cell (C2) has the value ("YES"), we could use the IF function with this formula

`=IF (C2="YES", B2, 0)` :

	A	B	C	D
1	Bills	Amount	Paid	Amounts already paid (if YES)
2	XLP-838893	\$100.00	YES	\$100.00
3	XLP-838894	\$200.00	YES	\$200.00
4	XLP-838895	\$300.00		\$0.00
5	XLP-838896	\$400.00	YES	\$400.00
6	XLP-838897	\$500.00		\$0.00
7	XLP-838898	\$600.00	YES	\$600.00
8	XLP-838899	\$700.00	YES	\$700.00
9	XLP-838900	\$800.00		\$0.00
10	XLP-838901	\$900.00	YES	\$900.00
11	XLP-838902	\$1'000.00		\$0.00
12	XLP-838903	\$1'100.00	YES	\$1'100.00
13	Total :			\$4'000.00
14				

Our goal in this case is to create a function that can do this `=IF (C2 has a green background, B2, 0)` which we would write as follows : `=IF_GREEN (C2, B2)` :

	A	B	C	D
1	Bills	Amount	Paid	Amounts already paid (if YES)
2	XLP-838893	\$100.00		
3	XLP-838894	\$200.00		
4	XLP-838895	\$300.00		
5	XLP-838896	\$400.00		
6	XLP-838897	\$500.00		
7	XLP-838898	\$600.00		
8	XLP-838899	\$700.00		
9	XLP-838900	\$800.00		
10	XLP-838901	\$900.00		
11	XLP-838902	\$1'000.00		
12	XLP-838903	\$1'100.00		
13	Total :			\$0.00
14				

Develop the custom function :

Start by creating the function :

```
Function IF_GREEN(paid As Range, amount)
End Function
```

The arguments :

- `paid As Range` : the cell to test
- `amount` : the value to insert if TRUE

In this case, if the test is FALSE, the value will be 0 every time, so there's no reason to have an argument for that.

To test to see whether the color is correct, you can use a cell that contains the color that you want as a point of reference :

```
Function IF_GREEN(paid As Range, amount)
    green_color = Sheets("Sheet1").Range("K1").Interior.color
End Function
```


But to avoid having to use another cell for this purpose, we'll use the number of the color that we're interested in here :

```
Function IF_GREEN(paid As Range, amount)
    green_color = 5296274 'Green
End Function
```

To figure out a cell's background color number, select the cell and run this macro :

```
Sub test_color()
    MsgBox ActiveCell.Interior.color
End Sub
```

	A	B	C	D	E
1	Bills	Amount	Paid	Amounts already paid (if YES)	
2	XLP-838893	\$100.00			
3	XLP-838894	\$200.00			
4	XLP-838895	\$300.00			
5	XLP-838896	\$400.00			
6	XLP-838897	\$500.00			
7	XLP-838898	\$600.00			
8	XLP-838899	\$700.00			
9	XLP-838900	\$800.00			
10	XLP-838901	\$900.00			
11	XLP-838902	\$1'000.00			
12	XLP-838903	\$1'100.00			
13	Total :			\$0.00	
14					



Now all we have to do is test the color number using **If** :

```

Function IF_GREEN(paid As Range, amount)

    Application.Volatile

    green_color = 5296274 'Green

    If paid.Interior.color = green_color Then 'If TRUE
        IF_GREEN = amount
    Else 'If FALSE
        IF_GREEN = 0
    End If

End Function

```

Our function is now ready to be used.

Application.Volatile tells us that the application is volatile (which is the case for the IF function). This means it should be recalculated each time a value is changed. For example, if you modify any of the amounts (or any other cell, in fact), the function will be recalculated and the correct amount will be displayed.

However, changing the background color does not trigger this recalculation. To force recalculation of the values, you can select an empty cell and press "Delete" or add an "Refresh" button that will cause everything to be updated when it is clicked :

```

Sub refresh_macro()
    Application.Calculate
End Sub

```

UPPER				
	A	B	C	D
1	Bills	Amount	Paid	Amounts already paid (if YES)
2	XLP-838893	\$100.00		=IF_GREEN(C2,B2)
3	XLP-838894	\$200.00		\$200.00
4	XLP-838895	\$300.00		\$0.00
5	XLP-838896	\$400.00		\$400.00
6	XLP-838897	\$500.00		\$0.00
7	XLP-838898	\$600.00		\$600.00
8	XLP-838899	\$700.00		\$700.00
9	XLP-838900	\$800.00		\$0.00
10	XLP-838901	\$900.00		\$900.00
11	XLP-838902	\$1'000.00		\$0.00
12	XLP-838903	\$1'100.00		\$1'100.00
13	Total :			\$4'000.00
14				Refresh

Extras :

Here is another way to write the same function :

```

Function IF_GREEN(paid As Range, amount)
Application.Volatile
IF_GREEN = 0 'If FALSE
If paid.Interior.color = 5296274 Then IF_GREEN = amount 'If TRUE
End Function

```

Source file : [if_green_function.xls/xlsm](#)